# -------------------------------------------------
# Project created by QtCreator 2009-09-26T17:25:52
# -------------------------------------------------
TARGET = basqet
TEMPLATE = app
QT += xml
include(uitools/uitools.pri)
include(mainwindow/mainwindow.pri)
include(basketmanager/basketmanager.pri)
include(datamanager/datamanager.pri)
include(commandmanager/commandmanager.pri)
include(pageview/pageview.pri)
include(outlineview/outlineview.pri)
include(workers/workers.pri)
include(tagmanager/tagmanager.pri)
SOURCES += main.cpp \
    xmlstrings.cpp \
    settings.cpp
OTHER_FILES += xmlstrings.hh \
    settings.hh \
    xml-example1.xml \
    pageview/note-creation.txt \
    gpl-3.0.txt
HEADERS += xmlstrings.h \
    settings.h \
    releasetag.h
RESOURCES += resources/basqet.qrc
RC_FILE = basqet.rc

# Unix installer.
unix:!macx { 
    isEmpty(PREFIX):PREFIX = /usr/local
    isEmpty(BINDIR):BINDIR = $$PREFIX/bin
    isEmpty(DATAROOT):DATAROOT = $$PREFIX/share
    isEmpty(DESKTOPDIR):DESKTOPDIR = $$DATAROOT/applications
    isEmpty(IMAGEDIR):IMAGEDIR = $$DATAROOT/pixmaps
    target.path = $$BINDIR
    desktop.path = $$DESKTOPDIR
    desktop.files = basqet.desktop
    icons.path = $$IMAGEDIR
    icons.files = resources/basqet.png
    INSTALLS += target \
        desktop \
        icons
}

# macosx specific stuff
macx {
    TARGET = Basqet
    ICON = macosx/basqet.icns
    QMAKE_INFO_PLIST = macosx/Info.plist    
}

